


## Настройка параметров при построении графиков

При построении графика система Maple автоматически выбирает масштаб по оси ординат для того, чтобы отразить на рисунке все наиболее существенные особенности отображаемой функции.


```
> plot(sin(x)/(1+sqrt(x)), x=0..4*Pi);
```



Если необходимо отключить автомасштабирование, то можно в команде **plot** указать третий параметр, который должен представлять собой диапазон. Будут отображаться только те части графика, которые попадают в указанный диапазон.

Данная возможность существует во всех версиях Maple. Однако в последних версиях документации она не упоминается. Тем не менее, соответствующий пример можно найти на странице справки [plot/details](#), хотя и без комментариев.


```
> plot(sin(x)/(1+sqrt(x)), x=0..4*Pi, y=-0.2..0.2);
```



Более гибкий способ отображения нужных фрагментов графика предполагает


использование опции *view*. Значением этой опции является список из двух диапазонов. Первый элемент этого списка может переопределить второй параметр команды **plot**.

```
> plot(sin(x)/(1+sqrt(x)), x=0..4*Pi, view=[2*Pi..3*Pi, 0..0.3]);
```



В верхней части рисунка можно сделать заголовок. Для этого используется опция *title*. В простейшем случае значением этой опции является текстовая строка. По умолчанию заголовок у рисунка отсутствует.


```
> plot(sin(x)/(1+sqrt(x)), x=0..4*Pi, title="Это заголовок!");
```



Если заголовок длинный, то его можно разбить на несколько строк.


Разделитель строк представляет собой пару символов "\n" как в языке С.

```
> plot(sin(x)+cos(x), x=0..3*Pi, title="Сумма двух\nфункций");
```



Maple позволяет создавать заголовки, которые представляют собой смесь из текстовых строк и математических формул. Фрагменты такого заголовка должны быть аргументами команды **typeset**. Строки набираются как обычно в двойных кавычках. Математические выражения - в соответствии с синтаксисом языка системы Maple.

```
> plot(sin(x)+cos(x), x=0..3*Pi,
      title=typeset("Сумма функций\n", sin(x)+cos(x)));
```




Заголовки принято писать большими буквами. Для изменения шрифта, которым выводится заголовок, используется опция *titlefont*. Значением этой опции является список. Этот список может содержать до трёх параметров шрифта: **[гарнитура, начертание, размер]**.

Сам Maple поставляется с четырьмя шрифтами, которые могут быть использованы при любых обстоятельствах: **Times, Courier, Helvetica, Symbol**. Возможно использование и других шрифтов, поддерживаемых операционной системой. Переносимость документов при этом не гарантируется. Название шрифтовой гарнитуры должно записываться с большой буквы.

Начертание является необязательным параметром. У шрифта **Symbol** и вовсе нет такого параметра. Размер шрифта указывается в пунктах.

```
> plot(sin(x)+cos(x), x=0..3*Pi,
      title=typeset("Сумма функций\n", sin(x)+cos(x)),
      titlefont=["Times", "Roman", 16]);
```


Сумма функций  
 $\sin(x) + \cos(x)$



Линии, которыми строятся графики, хорошо смотрятся на экране. Однако при их распечатке на принтерах с высоким разрешением они могут оказаться чересчур тонкими. Для изменения толщины линии используйте опцию *thickness*. Значением этой опции является неотрицательное число. По умолчанию оно равно 1.


```
> plot(sin(x)+cos(x), x=0..3*Pi,
      title=typeset("Сумма функций\n", sin(x)+cos(x)),
      titlefont=["Times", "Roman", 16], thickness=3);
```

Сумма функций  
 $\sin(x) + \cos(x)$



При построении графиков разрывных функций на рисунке могут появляться асимптоты в виде вертикальных линий.


```
> plot(cos(x)/(1+sin(x)), x=0..4*Pi, y=-5..5);
```



Для их удаления следует использовать опцию *discont* со значением *true*.


Эта опция может существенно увеличить время построения графика.

```
> plot(cos(x) / (1 + sin(x)), x=0..4*Pi, y=-5..5, discont=true);
```




Её можно также использовать для "борьбы" с разрывами первого рода. По умолчанию значение этой опции равно *false*, поэтому на графике возникают ступеньки.

```
> plot(floor(x), x=0..6);
```




Если значение опции равно *true* или же представляет собой список, то вызывается специальная процедура поиска разрывов.

```
> plot(floor(x), x=0..6, discontinuity=true);
```



Правила хорошего тона предполагают, что координатные оси соответствующим образом подписаны. Для этого предназначена опция *labels*. В качестве значения можно указать список из двух строк.

```
> plot(sin(x)^3+cos(x)^3, x=0..2*Pi, labels=["аргумент", "функция"]);
```



Полезным свойством оказывается возможность использования команды **typeset** для задания меток.

```
> plot(abs(x-1)-abs(x-3), x=0..5,
      labels=["аргумент", typeset(abs(x-1)-abs(x-3))] );
```

